


8. Οδηγός μελέτης – Δομή Πολλαπλής Επιλογής (Θ Κεφ. 2.4.2, 2.4.3, 2.4.4, 8.1.1) (Ο 2.3, 2.4) (Π Κεφ. 3)

8.1 Εμφωλευμένη δομή επιλογής

Η γενική μορφή της εντολής AN καλύπτει την επιλογή μεταξύ δύο εναλλακτικών περιπτώσεων. Όταν οι εναλλακτικές περιπτώσεις είναι περισσότερες από δύο τότε μπορούμε να βάλουμε μια Δομή Επιλογής μέσα σε μία άλλη. Η εσωτερική Δομή Επιλογής, ονομάζεται **εμφωλευμένη**.

Παράδειγμα 1


Γράψτε ένα αλγόριθμο στον οποίο δίνουμε έναν αριθμό και εμφανίζει αν είναι θετικός, αρνητικός ή μηδέν.



Βλέπουμε το παράδειγμα λυμένο με πολλούς τρόπους. Στον **1^ο αλγόριθμο** χρησιμοποιούμε δύο **σύνθετες δομές επιλογής** την μία μέσα στην άλλη, Στον **2^ο αλγόριθμο** χρησιμοποιούμε την **δομή πολλαπλής επιλογής**, ο πλέον κατάλληλος τρόπος, και στον **3^ο αλγόριθμο** χρησιμοποιούμε 3 **απλές δομές επιλογής**, την μία μετά την άλλη. Αν και με τους τρεις αλγόριθμους θα βρούμε το σωστό αποτέλεσμα, ο 3^{ος} αλγόριθμος θεωρείται ο λιγότερο αποτελεσματικός αφού απαιτεί τον έλεγχο περισσότερων συνθηκών.

Παράδειγμα 2

Αν ο αριθμός μας είναι το 1 θέλουμε να εμφανίσουμε το μήνυμα «ΠΕΡΙΠΤΩΣΗ 1» και για κάθε άλλο θετικό αριθμό το μήνυμα «ΠΕΡΙΠΤΩΣΗ Ζ». Είναι τα δύο παρακάτω τμήματα σωστά; Αιτιολογήστε την απάντησή σας.



Η σειρά με την οποία βάζουμε τις περιπτώσεις μπορεί να επηρεάσει το αποτέλεσμα. Η σωστή πρακτική είναι να ξεκινάμε από τις πιο ειδικές περιπτώσεις και στην συνέχεια να επεκτεινόμαστε στις πιο γενικές.

8.2 Δομή πολλαπλής επιλογής

Η χρήση πολλών εμφωλευμένων εντολών AN οδηγεί συνήθως σε πολύπλοκες δομές που αυξάνουν την πιθανότητα του λάθους καθώς και τη δυσκολία κατανόησης του προγράμματος. Συχνά οι εντολές που έχουν γραφεί με εμφωλευμένα AN, μπορούν να γραφούν πιο απλά χρησιμοποιώντας την δομή πολλαπλής επιλογής **ΑΝ ... ΤΟΤΕ ... ΑΛΛΙΩΣ_ΑΝ ...**

Παράδειγμα 3

Ένα κατάστημα πουλάει τους υπολογιστές σύμφωνα με τον διπλανό πίνακα. Γράψτε ένα τμήμα που διαβάζει τα τεμάχια αγοράς και υπολογίζει το κόστος.

ΠΟΣΟΤΗΤΑ	ΤΙΜΗ ΜΟΝΑΔΑΣ
1-50	580
51-100	520
101-200	470
πάνω από 200	440

Εμφωλευμένες δομές επιλογής	Δομή πολλαπλής επιλογής
<pre> ΔΙΑΒΑΣΕ Π ΑΝ Π <= 50 ΤΟΤΕ Κ <- Π*580 ΑΛΛΙΩΣ ΑΝ Π <= 100 ΤΟΤΕ Κ <- Π*520 ΑΛΛΙΩΣ ΑΝ Π <= 200 ΤΟΤΕ Κ <- Π*470 ΑΛΛΙΩΣ Κ <- Π*440 ΤΕΛΟΣ_ΑΝ ΤΕΛΟΣ_ΑΝ ΤΕΛΟΣ_ΑΝ ΓΡΑΨΕ Κ </pre>	<pre> ΔΙΑΒΑΣΕ Π ΑΝ Π <= 50 ΤΟΤΕ Κ <-- Π * 580 ΑΛΛΙΩΣ_ΑΝ Π <= 100 ΤΟΤΕ Κ <-- Π * 520 ΑΛΛΙΩΣ_ΑΝ Π <= 200 ΤΟΤΕ Κ <-- Π * 470 ΑΛΛΙΩΣ Κ <-- Π * 440 ΤΕΛΟΣ_ΑΝ ΓΡΑΨΕ Κ </pre>



Στην δομή πολλαπλής επιλογής ελέγχουμε την επόμενη περίπτωση μόνο όταν δεν αληθεύει η προηγούμενη! Άρα ο έλεγχος ότι δεν αληθεύει η προηγούμενη είναι περιττός.

Παράδειγμα 4

Να ξαναγραφεί ο αλγόριθμος του παραδείγματος 4 χρησιμοποιώντας αποκλειστικά και μόνο μη εμφωλευμένες **απλές δομές επιλογής**. Στην μετατροπή δεν πρέπει να ξεχνάμε να προσθέτουμε σε κάθε συνθήκη την πληροφορία που μεταβιβάζει το αλλιώς στην δομή πολλαπλής επιλογής.

```

ΔΙΑΒΑΣΕ Π
ΑΝ Π <= 50 ΤΟΤΕ
    Κ <- Π * 580
ΤΕΛΟΣ_ΑΝ
ΑΝ Π > 50 ΚΑΙ Π <= 100 ΤΟΤΕ
    Κ <- Π * 520
ΤΕΛΟΣ_ΑΝ
ΑΝ Π > 100 ΚΑΙ Π <= 200 ΤΟΤΕ
    Κ <- Π * 470
ΤΕΛΟΣ_ΑΝ
ΑΝ Π > 200 ΤΟΤΕ
    Κ <- Π * 440
ΤΕΛΟΣ_ΑΝ
ΓΡΑΨΕ Κ
    
```

Παράδειγμα 5 (Πανελλήνιες)

Μια εντολή εξόδου στο παραπάνω τμήμα δεν πρόκειται να εκτελεστεί, όποια και αν είναι η τιμή του Χ. Ποια είναι η εντολή αυτή; Να γράψετε τις εντολές εξόδου που είναι δυνατόν να εκτελεστούν και, δίπλα σε καθεμία από αυτές, το διάστημα τιμών του Χ για το οποίο θα εκτελεστεί η εντολή.

```

Διάβασε Χ
Αν Χ > 15 τότε
    Γράψε 1
αλλιώς αν Χ > 23 τότε
    Γράψε 2
αλλιώς
    Γράψε 3
Τέλος_αν
    
```

(-∞, 15] ΓΡΑΨΕ 3
(15, ∞) ΓΡΑΨΕ 1



Και στην δομή πολλαπλής επιλογή πρέπει να δίνουμε ιδιαίτερη προσοχή στην σειρά που ελέγχουμε τις συνθήκες!

8.3 Δομή επιλογής αλγόριθμοι

Παράδειγμα 6

Γράψτε τμήμα αλγόριθμου που υπολογίζει το αποτέλεσμα της παρακάτω συνάρτησης

$$\Phi(X) = \begin{cases} X + 12 & \text{αν } X = 0 \\ X^2 + 5 \cdot X & \text{αν } X \text{ άρτιος} \\ X / 7 & \text{αν } X \text{ περιττός} \end{cases}$$

```
ΔΙΑΒΑΣΕ X
ΑΝ X = 0 ΤΟΤΕ
  ΑΠ <-- X + 12
ΓΡΑΨΕ ΑΠ
ΑΛΛΙΩΣ
  ΑΝ Xmod2=0 ΤΟΤΕ
    ΑΠ <-- X^2 + 5*X
    ΓΡΑΨΕ ΑΠ
  ΑΛΛΙΩΣ
    ΑΠ <-- X / 7
    ΓΡΑΨΕ ΑΠ
ΤΕΛΟΣ_ΑΝ
ΤΕΛΟΣ_ΑΝ
```

Ερώτηση

Μπορείτε να αλλάξετε τον αλγόριθμο στο παράδειγμα 3 ώστε η εντολή «ΓΡΑΨΕ ΑΠ» να γραφτεί μόνο μια φορά μέσα στον αλγόριθμο; **ΝΑΙ**

Παράδειγμα 7

Ο Δείκτης Μάζας του ανθρώπινου Σώματος (ΔΜΣ) υπολογίζεται από το βάρος (Β) σε χλγ. και το ύψος (Υ) σε μέτρα με τον τύπο $\Delta\text{Μ}\Sigma = B/Y^2$. Ο ανωτέρω τύπος ισχύει για άτομα άνω των 18 ετών. Το άτομο ανάλογα με την τιμή του ΔΜΣ χαρακτηρίζεται σύμφωνα με τον παρακάτω πίνακα:

Να γράψετε πρόγραμμα το οποίο να διαβάζει την ηλικία, το βάρος και το ύψος του ατόμου και εάν η ηλικία είναι μεγαλύτερη ίση των 18 ετών, τότε να υπολογίζει το ΔΜΣ και να εμφανίζει τον αντίστοιχο χαρακτηρισμό αλλιώς εάν η ηλικία είναι μικρότερη των 18, τότε να εμφανίζει το μήνυμα «δεν ισχύει ο ΔΜΣ»

```
ΔΙΑΒΑΣΕ ΗΛ, Β, Υ
ΑΝ ΗΛ >= 18 ΤΟΤΕ
  ΔΜΣ <-- B/Y^2
  ΑΝ ΔΜΣ < 18.5 ΤΟΤΕ
    ΓΡΑΨΕ 'ΑΔΥΝΑΤΟ'
  ΑΛΛΙΩΣ_ΑΝ ΔΜΣ < 25 ΤΟΤΕ
    ΓΡΑΨΕ 'ΚΑΝΟΝΙΚΟ'
  ΑΛΛΙΩΣ_ΑΝ ΔΜΣ < 30 ΤΟΤΕ
    ΓΡΑΨΕ 'ΒΑΡΥ'
  ΑΛΛΙΩΣ
    ΓΡΑΨΕ 'ΥΠΕΡΒΑΡΟ'
  ΤΕΛΟΣ_ΑΝ
ΑΛΛΙΩΣ
  ΓΡΑΨΕ 'ΔΕΝ ΙΣΧΥΕΙ Ο ΔΜΣ'
ΤΕΛΟΣ_ΑΝ
```

Εμφωλευμένη Δομή
πολ/πλης Επιλογής



Συστήνεται να χρησιμοποιούμε μόνο μία δομή πολλαπλής επιλογής μόνο όταν ελέγχουμε τις περιπτώσεις ενός γεγονότος. Σε διαφορετική περίπτωση να χρησιμοποιούμε εμφωλευμένες δομές.

8.4 Δομή πολλαπλής επιλογής «Επίλεξε»

Παράδειγμα 4 – Σελίδα 3

```
ΠΡΟΓΡΑΜΜΑ ΑΣΚ1
ΜΕΤΑΒΛΗΤΕΣ
  ΠΡΑΓΜΑΤΙΚΕΣ: Π, Κ
ΑΡΧΗ
  ΔΙΑΒΑΣΕ Π
  ΑΝ Π <= 50 ΤΟΤΕ
    Κ <-- Π * 580
  ΑΛΛΙΩΣ_ΑΝ Π <= 100 ΤΟΤΕ
    Κ <-- Π * 520
  ΑΛΛΙΩΣ_ΑΝ Π <= 200 ΤΟΤΕ
    Κ <-- Π * 470
  ΑΛΛΙΩΣ
    Κ <-- Π * 440
  ΤΕΛΟΣ_ΑΝ
  ΓΡΑΨΕ Κ
ΤΕΛΟΣ_ΠΡΟΓΡΑΜΜΑΤΟΣ
```

```
ΠΡΟΓΡΑΜΜΑ ΑΣΚ1
ΜΕΤΑΒΛΗΤΕΣ
  ΠΡΑΓΜΑΤΙΚΕΣ: Π, Κ
ΑΡΧΗ
  ΔΙΑΒΑΣΕ Π
  Επίλεξε Π
    ΠΕΡΙΠΤΩΣΗ <= 50
      Κ <-- Π * 580
    ΠΕΡΙΠΤΩΣΗ <= 100
      Κ <-- Π * 520
    ΠΕΡΙΠΤΩΣΗ <= 200
      Κ <-- Π * 470
    ΠΕΡΙΠΤΩΣΗ ΑΛΛΙΩΣ
      Κ <-- Π * 440
  ΤΕΛΟΣ_ΕΠΙΛΟΓΩΝ
  ΓΡΑΨΕ Κ
ΤΕΛΟΣ_ΠΡΟΓΡΑΜΜΑΤΟΣ
```



ΠΕΡΙΠΤΩΣΗ λίστα_τιμών

ΠΕΡΙΠΤΩΣΗ 5	$X = 5$
ΠΕΡΙΠΤΩΣΗ 5, 7, 8	$X = 5 \text{ Η } X = 7 \text{ Η } X = 8$
ΠΕΡΙΠΤΩΣΗ 'Α', 'Β'	$X = 'Α' \text{ Η } X = 'Β'$
ΠΕΡΙΠΤΩΣΗ ≤ 2000	$X \leq 2000$
ΠΕΡΙΠΤΩΣΗ 0..2000	$X \geq 0 \text{ ΚΑΙ } X \leq 2000$



Το διάγραμμα ροής όταν έχουμε της δομής **ΕΠΙΛΕΞΕ** είναι ακριβώς το ίδιο με αυτό που θα κάναμε αν χρησιμοποιούσαμε την δομή πολλαπλής επιλογής **ΑΝ ... ΑΛΛΙΩΣ_ΑΝ**

8.5 Μετατροπές

Προσθήκη – Αφαίρεση λογικών πράξεων

Όταν έχουμε σε ένα πρόβλημα δύο ή περισσότερες περιπτώσεις που χρήζουν ακριβώς την ίδια αντιμετώπιση η χρήση μιας **σύνθετης λογικής έκφρασης** μπορεί να απλοποιήσει το πρόγραμμά μας όπως στο παρακάτω παράδειγμα.

```
ΔΙΑΒΑΣΕ ΒΑΡΟΣ, ΥΨ
ΑΝ ΒΑΡΟΣ < 80 ΤΟΤΕ
  ΑΝ ΥΨ < 1.70 ΤΟΤΕ
    ΓΡΑΨΕ 'ΕΛΑΦΡΥΣ, ΚΟΝΤΟΣ'
  ΤΕΛΟΣ_ΑΝ
ΤΕΛΟΣ_ΑΝ
```

```
ΔΙΑΒΑΣΕ ΒΑΡΟΣ, ΥΨ
ΑΝ ΒΑΡΟΣ < 80 ΚΑΙ ΥΨ < 1.70 ΤΟΤΕ
  ΓΡΑΨΕ 'ΕΛΑΦΡΥΣ, ΚΟΝΤΟΣ'
ΤΕΛΟΣ_ΑΝ
```

```
ΑΝ Χ=0 ΤΟΤΕ
  ΓΡΑΨΕ 'ΟΚ'
ΑΛΛΙΩΣ_ΑΝ Χ=1 ΤΟΤΕ
  ΓΡΑΨΕ 'ΟΚ'
ΤΕΛΟΣ_ΑΝ
```

```
ΑΝ Χ=0 Η Χ=1 ΤΟΤΕ
  ΓΡΑΨΕ 'ΟΚ'
ΤΕΛΟΣ_ΑΝ
```

Εμφωλευμένες σε πολλαπλής και αντίστροφα


Η μετατροπή είναι απλή αρκεί να προσέχουμε να κλείσουμε όλες τις εμφωλευμένες δομές.

```
ΑΝ Α>0 ΤΟΤΕ
  ΓΡΑΨΕ 'ΘΕΤΙΚΟΣ'
ΑΛΛΙΩΣ
  ΑΝ Α<0 ΤΟΤΕ
    ΓΡΑΨΕ 'ΑΡΝΗΤΙΚΟΣ'
  ΑΛΛΙΩΣ
    ΓΡΑΨΕ 'ΜΗΔΕΝ'
  ΤΕΛΟΣ_ΑΝ
ΤΕΛΟΣ_ΑΝ
```

```
ΑΝ Α>0 ΤΟΤΕ
  ΓΡΑΨΕ 'ΘΕΤΙΚΟΣ'
ΑΛΛΙΩΣ_ΑΝ Α<0 ΤΟΤΕ
  ΓΡΑΨΕ 'ΑΡΝΗΤΙΚΟΣ'
ΑΛΛΙΩΣ
  ΓΡΑΨΕ 'ΜΗΔΕΝ'
ΤΕΛΟΣ_ΑΝ
```


Εμφωλευμένες σε απλές δομές επιλογής

Στην δομή πολλαπλής επιλογής μόνο η πρώτη ταιριαστή περίπτωση εκτελείται, και οι υπόλοιπες αγνοούνται αφού στο αλλιώς συνεχίζω μόνο αν οι προηγούμενες συνθήκες δεν ισχύουν. Αν πας να τη μετατρέψεις σε πολλές απλές ΑΝ, πρέπει να διατηρήσεις αυτή τη συμπεριφορά.




Απλές δομές επιλογής σε πολλαπλής επιλογής

Πριν τη μετατροπή, ο μαθητής πρέπει να αναλύσει αν οι απλές ΑΝ εκτελούνται περισσότερες από μία ΑΝ για την ίδια τιμή; Αν ναι, τότε δεν μπορεί να μετατραπεί αυτούσια αφού η δομή πολλαπλής επιλογής εκτελεί μόνο μία περίπτωση. Οπότε πρέπει να βρούμε όλα τα διαστήματα τιμών και για κάθε διάστημα θα πρέπει να αναζητήσουμε ποιες απλές δομές επιλογής θα εκτελεστούν και με την ίδια σειρά να τις βάλουμε στην δομή πολλαπλής επιλογής.



Σύνθετη δομή επιλογής σε μία απλή δομή επιλογής

Για να μπορεί να γίνει αυτή η μετατροπή θα πρέπει μια από τις δύο περιπτώσεις να περιέχει μόνο εντολές που είναι κοινές σε κάθε ενδεχόμενο. Οπότε εντολές που εκτελούνται σε όλες τις περιπτώσεις μπορεί να βγουν εκτός δομής επιλογής.



8.6 Κλιμακωτή χρέωση

Παράδειγμα 8

Η ΔΕΗ αγοράζει την ενέργεια από τους κατόχους των φωτοβολταϊκών πάρκων σύμφωνα με την παρακάτω ετήσια κλιμακωτή τιμολογιακή πολιτική :

KW	Κόστος / KW
1-1000	0,05€
1001-3000	0,06€
3001+	0,07€

Φτιάξτε ένα πρόγραμμα το οποίο θα διαβάζει την ετήσια παραγωγή σε KWh και θα εμφανίζει το ποσό που θα μας δώσει η ΔΕΗ στο τέλος του έτους με βάση τον παραπάνω πίνακα κλιμακωτής χρέωσης.



Στην **κλιμακωτή χρέωση** το κόστος διαμοιράζεται σε όλες τις κλίμακες.

Δηλαδή, με βάση τον παραπάνω πίνακα, ανεξαρτήτως πόσες Kwh έχουμε πάντα οι πρώτες 1000 θα χρεώνονται με 0,05€. Η επόμενη χρέωση με 0,06€ θα εφαρμόζεται μόνο για τις μονάδες που θα είναι πάνω από 1000 και μέχρι τις 3000 κ.τ.λ.

```

ΠΡΟΓΡΑΜΜΑ  ΦΩΤ

ΜΕΤΑΒΛΗΤΕΣ
  ΑΚΕΡΑΙΕΣ  : Εν
  ΠΡΑΓΜΑΤΙΚΕΣ : ΕΙΣ

ΑΡΧΗ
  ΔΙΑΒΑΣΕ  Εν

  ΑΝ Εν<=1000 ΤΟΤΕ
    ΕΙΣ <--  Εν * 0.05
  ΑΛΛΙΩΣ_ΑΝ Εν<=3000 ΤΟΤΕ
    ΕΙΣ <--  1000*0.05 + (Εν-1000)*0.06
  ΑΛΛΙΩΣ
    ΕΙΣ <--  1000*0.05 + 2000*0.06 + (Εν-3000)*0.07
  ΤΕΛΟΣ_ΑΝ

  ΓΡΑΨΕ 'Το ετήσιο κέρδος μας θα είναι ', ΕΙΣ
ΤΕΛΟΣ_ΠΡΟΓΡΑΜΜΑΤΟΣ  ΦΩΤ

```

Ακολουθεί αναλυτικά η μεθοδολογία που ακολουθούμε για την επίλυση των ασκήσεων που ζητούν κλιμακωτή χρέωση.



Μεθοδολογία κλιμακωτής χρέωσης

Έστω οι μονάδες που έχουμε διαβάσει είναι αποθηκευμένες στην μεταβλητή **Μονάδες** τότε με βάση τον πίνακα κλιμακωτής χρέωσης της άσκησης για **κάθε γραμμή** του πίνακα παίρνουμε και μια ξεχωριστή περίπτωση. **Επίσης χρήσιμο είναι να έχουμε υπολογίσει το διάστημα κάθε γραμμής πόσες Μονάδες περιέχει!**

Περίπτωση 1^{ης} γραμμής : 1 – 1000 (το διάστημα περιέχει 1000 Μονάδες)

Αφού δεν έχουμε ξεπεράσει τις 1000 μονάδες: **Μονάδες < 1000**

$$\text{Κόστος} = \text{Μονάδες} * 0,05$$

Όσες μονάδες και να έχουμε αφού είναι λιγότερες από 1000 αυτές θα χρεώνονται με 0,05€ η μία.

Περίπτωση 2^{ης} γραμμής : 1001 – 3000 (το διάστημα περιέχει 2000 Μονάδες)

Όταν οι μονάδες είναι μεταξύ : **Μονάδες > 1000 ΚΑΙ Μονάδες ≤ 3000**

$$\text{Κόστος} = 1000 * 0,05 + (\text{Μονάδες} - 1000) * 0,06$$

Οι πρώτες 1000 Μονάδες που περιέχει το πρώτο διάστημα χρεώνονται με το κόστος της 1^{ης} γραμμής του πίνακα **1000*0,05** και (+) όσες επιπλέον έχουμε δηλαδή **(Μονάδες - 1000)** με το κόστος της 2^{ης} γραμμής.

Περίπτωση 3^{ης} γραμμής : 3000+ (όλες οι μονάδες πάνω από 3000)

Όταν η μονάδες είναι πάνω από 3000 : **Μονάδες > 3000**

$$\text{Κόστος} = 1000*0,05 + 2000*0,06 + (\text{Μονάδες} - 3000) * 0,07$$

Οι πρώτες 1000 (πρώτο διάστημα) χρεώνονται με την 1^η γραμμή του πίνακα **1000*0,05** + οι επόμενες 2000 (δεύτερο διάστημα) χρεώνονται με το κόστος της 2^{ης} γραμμής του πίνακα **2000*0,06** + επιπλέον όσες πάνω από τις 3000 έχουμε: **(Μονάδες - 3000) * 0,07** χρεώνονται με το κόστος της 3^{ης} γραμμής.

Γιατί **(Μονάδες - 3000)** ; Παρατηρούμε στην αριθμητική έκφραση της 3^{ης} περίπτωσης ότι για τις πρώτες 1000 και τις επόμενες 2000 Μονάδες έχουμε ήδη υπολογίσει το κόστος τους. Άρα μόνο όσες Μονάδες έχουμε πάνω από τις 3000 θα πρέπει να χρεώσουμε με 0,07€.